PROJET TUTORÉ N°2 - Arduino Temperature Data Logger with SD Card Module

M1S2 RIoT-IE, Dept. Informatique, U-JKZ

- Internet of Things (IoT) -

- Arduino, SD Card Module -

Projet à rendre au plus tard le 30 Avril 2026 à l'adresse mail suivante :
desire.guel@ujkz.bf

Instructions

— Ce projet comprend différentes parties dépendantes.

— Il est à mener par groupe de trois (3) à quatre (4) personnes tout au plus.

— Le travail doit être rendu en format pdf au plutard le **30 Avril 2026** à l'adresse mail suivante : **desire.guel@ujkz.bf**

— Le dossier zip devra contenir les éléments suivants :

   (a) Documentation : Produisez une documentation complète expliquant le fonctionnement du projet, le schéma de connexion, et le code source.

   (b) Les fichiers sources du projet en lien avec la deuxième partie du projet

— Une séance de démonstration est prévue dont la date sera communiquée ultérieurement.

Introduction

Ce projet vise à vous guider dans la création d'un enregistreur de données de température utilisant Arduino. Il offre une opportunité d'application pratique des compétences en ingénierie, programmation et conception de systèmes embarqués, tout en développant un enregistreur de données de température fonctionnel avec des composants Arduino. Pour ce faire, vous alles utiliser le capteur DHT11 pour mesurer la température, le module d'horloge temps réel (RTC) pour estampiller les données temporelles, et le module de carte SD pour sauvegarder les informations sur la carte SD.

Instructions détaillées

- Capteur de Température : Connectez le capteur DHT11 à Arduino pour mesurer avec précision la température ambiante.
- Module RTC : Intégrez le module RTC pour capturer les horodatages des données, assurant une chronologie précise des mesures.
- Module Carte SD : Connectez le module de carte SD à Arduino pour créer un système d'enregistrement robuste.
- Programmation : Utilisez le langage de programmation Arduino pour développer un code efficace gérant la collecte de données et leur sauvegarde sur la carte SD.
- Tests et Validation : Effectuez des tests approfondis pour vous assurer que le dispositif fonctionne correctement, enregistrant avec succès les données de température sur la carte SD.

Arduino Temperature Data Logger with SD Card Module



View Project on Random Nerd Tutorials	Click [here]()
View code on GitHub	Click [here]()

This project shows you how to create an Arduino temperature data logger. We'll use the DHT11 to measure temperature, the real time clock (RTC) module to take time stamps and the SD card module to save the data on the SD card.

Recommended resources:

- Guide for Real Time Clock (RTC) Module with Arduino (DS1307 and DS3231)
- Complete Guide for DHT11/DHT22 Humidity and Temperature Sensor With Arduino
- Guide to SD Card module with Arduino

Parts Required

Here's a complete list of the parts required for this project:



- Arduino UNO – read Best Arduino Starter Kits
- SD card module
- Micro SD card
- DHT11 temperature and humidity sensor
- RTC module
- Breadboard
- Jumper wires

Note: alternatively to the SD card module, you can use a data logging shield. The data logging shield comes with built-in RTC and a prototyping area for soldering connections, sensors, etc..

Schematics

The following figure shows the circuit's schematics for this project.



Note: make sure your SD card is formatted and working properly. Read "Guide to SD card module with Arduino".

Installing the DHT Sensor Library

For this project you need to install the DHT library to read from the DHT11 sensor.

1. Click here to download the DHT-sensor-library. You should have a .zip folder in your Downloads folder
2. Unzip the .zip folder and you should get **DHT-sensor-library-master** folder
3. Rename your folder from **~~DHT-sensor-library-master~~** to **DHT**
4. Move the **DHT** folder to your Arduino IDE installation libraries folder
5. Finally, re-open your Arduino IDE

Installing the Adafruit_Sensor library

To use the DHT temperature and humidity sensor, you also need to install the Adafruit_Sensor library. Follow the next steps to install the library in your Arduino IDE:

1. Click here to download the Adafruit_Sensor library. You should have a .zip folder in your Downloads folder
2. Unzip the .zip folder and you should get **Adafruit_Sensor-master** folder
3. Rename your folder from **~~Adafruit_Sensor-master~~** to **Adafruit_Sensor**

4. Move the **Adafruit_Sensor** folder to your Arduino IDE installation libraries folder
5. Finally, re-open your Arduino IDE

Code

Copy the following code to your Arduino IDE and upload it to your Arduino board.

[View code on GitHub]

```
/*
 * Rui Santos
 * Complete Project Details http://randomnerdtutorials.com
 */

#include <SPI.h> //for the SD card module
#include <SD.h> // for the SD card
#include <DHT.h> // for the DHT sensor
#include <RTClib.h> // for the RTC

//define DHT pin
#define DHTPIN 2     // what pin we're connected to

// uncomment whatever type you're using
#define DHTTYPE DHT11   // DHT 11
//#define DHTTYPE DHT22   // DHT 22  (AM2302)
//#define DHTTYPE DHT21   // DHT 21 (AM2301)

// initialize DHT sensor for normal 16mhz Arduino
DHT dht(DHTPIN, DHTTYPE);

// change this to match your SD shield or module;
// Arduino Ethernet shield and modules: pin 4
// Data loggin SD shields and modules: pin 10
// Sparkfun SD shield: pin 8
const int chipSelect = 4;

// Create a file to store the data
File myFile;

// RTC
RTC_DS1307 rtc;

void setup() {
  //initializing the DHT sensor
  dht.begin();

  //initializing Serial monitor
  Serial.begin(9600);
```

```
  // setup for the RTC
  while(!Serial); // for Leonardo/Micro/Zero
    if(! rtc.begin()) {
      Serial.println("Couldn't find RTC");
      while (1);
    }
    else {
      // following line sets the RTC to the date & time this sketch was
compiled
      rtc.adjust(DateTime(F(__DATE__), F(__TIME__)));
    }
    if(! rtc.isrunning()) {
      Serial.println("RTC is NOT running!");
    }

  // setup for the SD card
  Serial.print("Initializing SD card...");

  if(!SD.begin(chipSelect)) {
    Serial.println("initialization failed!");
    return;
  }
  Serial.println("initialization done.");

  //open file
  myFile=SD.open("DATA.txt", FILE_WRITE);

  // if the file opened ok, write to it:
  if (myFile) {
    Serial.println("File opened ok");
    // print the headings for our data
    myFile.println("Date,Time,Temperature °C");
  }
  myFile.close();
}

void loggingTime() {
  DateTime now = rtc.now();
  myFile = SD.open("DATA.txt", FILE_WRITE);
  if (myFile) {
    myFile.print(now.year(), DEC);
    myFile.print('/');
    myFile.print(now.month(), DEC);
    myFile.print('/');
    myFile.print(now.day(), DEC);
    myFile.print(',');
    myFile.print(now.hour(), DEC);
    myFile.print(':');
    myFile.print(now.minute(), DEC);
    myFile.print(':');
    myFile.print(now.second(), DEC);
    myFile.print(",");
```

```
  }
  Serial.print(now.year(), DEC);
  Serial.print('/');
  Serial.print(now.month(), DEC);
  Serial.print('/');
  Serial.println(now.day(), DEC);
  Serial.print(now.hour(), DEC);
  Serial.print(':');
  Serial.print(now.minute(), DEC);
  Serial.print(':');
  Serial.println(now.second(), DEC);
  myFile.close();
  delay(1000);
}

void loggingTemperature() {
  // Reading temperature or humidity takes about 250 milliseconds!
  // Sensor readings may also be up to 2 seconds 'old' (its a very slow
sensor)
  // Read temperature as Celsius
  float t = dht.readTemperature();
  // Read temperature as Fahrenheit
  //float f = dht.readTemperature(true);

  // Check if any reads failed and exit early (to try again).
  if  (isnan(t) /*|| isnan(f)*/) {
    Serial.println("Failed to read from DHT sensor!");
    return;
  }

  //debugging purposes
  Serial.print("Temperature: ");
  Serial.print(t);
  Serial.println(" *C");
  //Serial.print(f);
  //Serial.println(" *F\t");

  myFile = SD.open("DATA.txt", FILE_WRITE);
  if (myFile) {
    Serial.println("open with success");
    myFile.print(t);
    myFile.println(",");
  }
  myFile.close();
}

void loop() {
  loggingTime();
  loggingTemperature();
  delay(5000);
}
```

In this code we create two functions that are called in the loop():

- **loggingTime()**: log time to the **DATA.txt** file in the SD card
- **loggingTemperature()**: log temperature to the **DATA.txt** file in the SD card.

Open the Serial Monitor at a baud rate of 9600 and check if everything is working properly.

Getting the Data from the SD Card

Let this project run for a few hours to gather a decent amount of data, and when you're happy with the data logging period, shut down the Arduino and remove the SD from the SD card module.

Insert the SD card on your computer, open it, and you should have a **DATA.txt** file with the collected data.

You can open the data with a text editor, or use a spreadsheet to analyse and manipulate your data.




```
Date,Time,Temperature ºC
2017/8/24,19:3:26,25.40,
2017/8/24,19:3:32,25.40,
2017/8/24,19:3:39,25.40,
2017/8/24,19:3:45,25.40,
2017/8/24,19:3:51,25.30,
2017/8/24,19:3:58,25.30,
2017/8/24,19:4:4,25.30,
2017/8/24,19:4:10,25.30,
2017/8/24,19:4:17,25.30,
2017/8/24,19:4:23,25.30,
2017/8/24,19:4:29,25.30,
2017/8/24,19:4:36,25.30,
2017/8/24,19:4:42,25.30,
2017/8/24,19:4:48,25.30,
2017/8/24,19:4:55,25.30,
2017/8/24,19:5:1,25.30,
2017/8/24,19:5:7,25.30,
2017/8/24,19:5:14,25.30,
2017/8/24,19:5:20,25.30,
2017/8/24,19:5:26,25.20,
2017/8/24,19:5:32,25.30,
2017/8/24,19:5:39,25.20,
2017/8/24,19:5:45,25.20,
2017/8/24,19:5:51,25.20,
2017/8/24,19:5:58,25.20,
```

The data is separated by commas, and each reading is in a new line. In this format, you can easily import data to Excel or other data processing software.